lESSON nO.33

BIOS Video Services

The Basic Input Output System (BIOS) provides services for video, keyboard, serial port, parallel port, time etc. The video services are exported via INT 10. We will discuss some very simple services. Video services are classified into two broad categories; graphics mode services and text mode services. In graphics mode a location in video memory corresponds to a dot on the screen. In text mode this relation is not straightforward. The video memory holds the ASCII of the character to be shown and the actual shape is read from a font definition stored elsewhere in memory. We first present a list of common video services used in text mode.

INT 10 - VIDEO - SET VIDEO MODE

AH = 00h

AL = desired video mode 
Some common video modes include 40x25 text mode (mode 0), 80x25 text mode (mode 2), 80x50 text mode (mode 3), and 320x200 graphics mode (mode D). 

INT 10 - VIDEO - SET TEXT-MODE CURSOR SHAPE

AH = 01h

CH = cursor start and options
CL = bottom scan line containing cursor (bits 0-4)

INT 10 - VIDEO - SET CURSOR POSITION

AH = 02h

BH = page number

    0-3 in modes 2&3

    0-7 in modes 0&1

    0 in graphics modes

DH = row (00h is top)

DL = column (00h is left)

INT 10 - VIDEO - SCROLL UP WINDOW

AH = 06h

AL = number of lines by which to scroll up (00h = clear entire window)

BH = attribute used to write blank lines at bottom of window

CH, CL = row, column of window's upper left corner

DH, DL = row, column of window's lower right corner

INT 10 - VIDEO - SCROLL DOWN WINDOW

AH = 07h

AL = number of lines by which to scroll down (00h=clear entire window)

BH = attribute used to write blank lines at top of window

CH, CL = row, column of window's upper left corner

DH, DL = row, column of window's lower right corner

INT 10 - VIDEO - WRITE CHARACTER AND ATTRIBUTE AT CURSOR POSITION

AH = 09h

AL = character to display

BH = page number
BL = attribute (text mode) or color (graphics mode)

CX = number of times to write character

INT 10 - VIDEO - WRITE CHARACTER ONLY AT CURSOR POSITION

AH = 0Ah

AL = character to display

BH = page number 
BL = attribute (text mode) or color (graphics mode)

CX = number of times to write character

INT 10 - VIDEO - WRITE STRING
AH = 13h

AL = write mode

   bit 0: update cursor after writing

   bit 1: string contains alternating characters and attributes

   bits 2-7: reserved (0)

BH = page number

BL = attribute if string contains only characters

CX = number of characters in string

DH, DL = row, column at which to start writing

ES:BP -> string to write
Chargen Services

In our first example we will read the font definition in memory and change it to include a set of all on pixels in the last line showing an effect of underline on all character including space. An 8x16 font is stored in 16 bytes. A sample character and the corresponding 16 values stored in the font information are shown for the character ‘A’. We start with two services from the chargen subset of video services that we are going to use.

INT 10 - VIDEO - GET FONT INFORMATION
AX = 1130h

BH = pointer specifier

Return: 
ES:BP = specified pointer

CX = bytes/character of on-screen font 
DL = highest character row on screen

INT 10 - TEXT-MODE CHARGEN

AX = 1110h

ES:BP -> user table

CX = count of patterns to store

DX = character offset into map 2 block

BL = block to load in map 2

BH = number of bytes per character pattern

We will use 6 as the pointer specifier which means the 8x16 font stored in ROM.

	; put underlines on screen font
[org 0x0100]

              jmp  start

font:         times 256*16 db 0       ; space for font
start:        mov  ax, 0x1130         ; service 11/30 – get font info
              mov  bx, 0x0600         ; ROM 8x16 font
              int  0x10               ; bios video services
              mov  si, bp             ; point si to rom font data
              mov  di, font           ; point di to space for font
              mov  cx, 256*16         ; font size 
              push ds                 
              push es                 
              pop  ds                 ; ds:si to rom font data
              pop  es                 ; es:di to space for font
              cld                     ; auto increment mode
              rep  movsb              ; copy font
              push cs

              pop  ds                 ; restore ds to data segment
              mov  si, font-1         ; point si before first char
              mov  cx, 0x100          ; total 256 characters
change:       add  si, 16             ; one character has 16 bytes
              mov  byte [si], 0xFF    ; change last line to all ones
              loop change             ; repeat for each character
              mov  bp, font           ; es:bp points to new font
              mov  bx, 0x1000         ; bytes per char & block number
              mov  cx, 0x100          ; number of characters to change
              xor  dx, dx             ; first character to change
              mov  ax, 0x1110         ; service 11/10 – load user font
              int  0x10               ; bios video services
              mov ax, 0x4c00          ; terminate program
              int 0x21


Our second example is similar to the last example however in this case we are doing something funny on the screen. We are reversing the shapes of all the characters on the screen.

	; reverse each character of screen font
[org 0x0100]

              jmp  start

font:         times 256*16 db 0       ; space for font
start:        mov  ax, 0x1130         ; service 11/30 – get font info
              mov  bx, 0x0600         ; ROM 8x16 font
              int  0x10               ; bios video services
              mov  si, bp             ; point si to rom font data
              mov  di, font           ; point di to space for font
              mov  cx, 256*16         ; font size 
              push ds                 
              push es                 
              pop  ds                 ; ds:si to rom font data
              pop  es                 ; es:di to space for font
              cld                     ; auto increment mode
              rep  movsb              ; copy font
              push cs

              pop  ds                 ; restore ds to data segment
              mov si, font            ; point si to start of font
change:       mov al, [si]            ; read one byte
              mov cx, 8               
inner:        shl al, 1               ; shift left with MSB in carry
              rcr bl, 1               ; rotate right using carry
              loop inner              ; repeat eight times
              mov [si], bl            ; write back reversed byte
              inc si                  ; next byte of font
              cmp si, font+256*16     ; is whole font reversed
              jne change              ; no, reverse next byte

              mov  bp, font           ; es:bp points to new font
              mov  bx, 0x1000         ; bytes per char & block number
              mov  cx, 0x100          ; number of characters to change
              xor  dx, dx             ; first character to change
              mov  ax, 0x1110         ; service 11/10 – load user font
              int  0x10               ; bios video services
              mov  ax, 0x4c00         ; terminate program
              int  0x21


Graphics Mode Services

We will take an example of using graphics mode video services as well. We will draw a line across the screen using the following service.

INT 10 - VIDEO - WRITE GRAPHICS PIXEL

AH = 0Ch

BH = page number

AL = pixel color

CX = column

DX = row

	; draw line in graphics mode
[org 0x0100]

              mov  ax, 0x000D         ; set 320x200 graphics mode

              int  0x10               ; bios video services
              mov  ax, 0x0C07         ; put pixel in white color

              xor  bx, bx             ; page number 0
              mov  cx, 200            ; x position 200
              mov  dx, 200            ; y position 200
l1:           int  0x10               ; bios video services
              dec  dx                 ; decrease y position
              loop l1                 ; decrease x position and repeat
              mov  ah, 0              ; service 0 – get keystroke
              int  0x16               ; bios keyboard services
              mov  ax, 0x0003         ; 80x25 text mode

              int  0x10               ; bios video services

              mov  ax, 0x4c00         ; terminate program
              int  0x21


